# Meta−Groups, Meta−Bicrossed−Products, and the Alexander Polynomial, 2

A Meta-Bicrossed-Product is a collection of sets $\beta(\eta,\tau)$ and operations $tm^{uv}_w$, $hm^{xy}_z$ and $sw^{th}_{ux}$ (and lesser ones), such that $tm$ and $hm$ are "associative" and (1) and (2) hold (+ lesser conditions). A meta-bicrossed-product defines a meta-group with $G_\gamma := \beta(\gamma,\gamma)$ and $gm$ as in (3).

Example. Take $\beta(\eta,\tau) = M_{\tau\times\eta}(\mathbb{Z})$ with row operations for the tails, column operations for the heads, and a trivial swap.

$\beta$ Calculus. Let $\beta(\eta,\tau)$ be

$$\left\{\begin{array}{c|cccc} \omega & h_1 & h_2 & \cdots \\ \hline t_1 & \alpha_{11} & \alpha_{12} & \cdot \\ t_2 & \alpha_{21} & \alpha_{22} & \cdot \\ \vdots & \cdot & \cdot & \cdot \end{array} \;\middle|\; \begin{array}{l} h_j \in \eta,\ t_i \in \tau, \text{ and } \omega \text{ and} \\ \text{the } \alpha_{ij} \text{ are rational func-} \\ \text{tions in a variable } X \end{array}\right\},$$

$$tm^{uv}_w : \begin{array}{c|c} \omega & \cdots \\ \hline t_u & \alpha \\ t_v & \beta \\ \vdots & \gamma \end{array} \mapsto \begin{array}{c|c} \omega & \cdots \\ \hline t_w & \alpha+\beta \\ \vdots & \gamma \end{array}, \qquad \begin{array}{c|c} \omega_1 & \eta_1 \\ \hline \tau_1 & \alpha_1 \end{array} \cup \begin{array}{c|c} \omega_2 & \eta_2 \\ \hline \tau_2 & \alpha_2 \end{array} = \begin{array}{c|cc} \omega_1\omega_2 & \eta_1 & \eta_2 \\ \hline \tau_1 & \alpha_1 & 0 \\ \tau_2 & 0 & \alpha_2 \end{array},$$

$$hm^{xy}_z : \begin{array}{c|ccc} \omega & h_x & h_y & \cdots \\ \hline \vdots & \alpha & \beta & \gamma \end{array} \mapsto \begin{array}{c|cc} \omega & h_z & \cdots \\ \hline \vdots & \alpha+\beta+\langle\alpha\rangle\beta & \gamma \end{array},$$

$$sw^{th}_{ux} : \begin{array}{c|cc} \omega & h_x & \cdots \\ \hline t_u & \alpha & \beta \\ \vdots & \gamma & \delta \end{array} \mapsto \begin{array}{c|cc} \omega\epsilon & h_x & \cdots \\ \hline t_u & \alpha(1+\langle\gamma\rangle/\epsilon) & \beta(1+\langle\gamma\rangle/\epsilon) \\ \vdots & \gamma/\epsilon & \delta-\gamma\beta/\epsilon \end{array},$$

where $\epsilon := 1+\alpha$ and $\langle c\rangle := \sum_i c_i$, and let

$$R^p_{ab} := \begin{array}{c|cc} 1 & h_a & h_b \\ \hline t_a & 0 & X-1 \\ t_b & 0 & 0 \end{array} \qquad R^m_{ab} := \begin{array}{c|cc} 1 & h_a & h_b \\ \hline t_a & 0 & X^{-1}-1 \\ t_b & 0 & 0 \end{array}.$$

Theorem. $Z^\beta$ is a tangle invariant (and more). Restricted to knots, the $\omega$ part is the Alexander polynomial. On braids, it is equivalent to the Burau representation. A variant for links contains the multivariable Alexander polynomial.

Why Happy? • Applications to w-knots.
• Everything that I know about the Alexander polynomial can be expressed cleanly in this language (even if without proof), except HF, but including genus, ribbonness, cabling, v-knots, knotted graphs, etc., and there's potential for vast generalizations.
• The least wasteful "Alexander for tangles" I'm aware of.
• Every step along the computation is the invariant of something.
• Fits on one sheet, including implementation & propaganda.

Further meta-monoids. $\Pi$ (and variants), $\mathcal{A}$ (and quotients), $vT$, …

Further meta-bicrossed-products. $\Pi$ (and variants), $\overrightarrow{\mathcal{A}}$ (and quotients), $M_0$, $M$, $\mathcal{K}^{bh}$, $\mathcal{K}^{rbh}$, …

Meta-Lie-algebras. $\mathcal{A}$ (and quotients), $\mathcal{S}$, …

Meta-Lie-bialgebras. $\overrightarrow{\mathcal{A}}$ (and quotients), …

I don't understand the relationship between gr and $H$, as it appears, for example, in braid theory.

I mean business!

```
βSimp = Factor; SetAttributes[βCollect, Listable];
βCollect[B[ω_, Λ_]] := B[βSimp[ω],
   Collect[Λ, h_, Collect[#, t_, βSimp] &]];
βForm[B[ω_, Λ_]] := Module[{ts, hs, M},
   ts = Union[Cases[B[ω, Λ], t_u_ :> u, Infinity]];
   hs = Union[Cases[B[ω, Λ], h_x_ :> x, Infinity]];
   M = Outer[βSimp[Coefficient[Λ, h_#1 t_#2]] &, hs, ts];
   PrependTo[M, t_# & /@ ts];
   M = Prepend[Transpose[M], Prepend[h_# & /@ hs, ω]];
   MatrixForm[M]];
βForm[else_] := else /. β_B :> βForm[β];
Format[β_B, StandardForm] := βForm[β];
⟨μ_⟩ := μ /. t_ → 1;
tm_{u_ v_→w_}[β_] := βCollect[β /. t_{u|v} → t_w];
hm_{x_ y_→z_}[B[ω_, Λ_]] := Module[
   {α = D[Λ, h_x], β = D[Λ, h_y], γ = Λ /. h_{x|y} → 0},
   B[ω, (α + (1 + ⟨α⟩) β) h_z + γ] // βCollect];
sw_{u_ x_}[B[ω_, Λ_]] := Module[{α, β, γ, δ, ϵ},
   α = Coefficient[Λ, h_x t_u]; β = D[Λ, t_u] /. h_x → 0;
   γ = D[Λ, h_x] /. t_u → 0;   δ = Λ /. h_x | t_u → 0;
   ϵ = 1 + α;
   B[ω * ϵ, α (1 + ⟨γ⟩ / ϵ) h_x t_u + β (1 + ⟨γ⟩ / ϵ) t_u
        + γ / ϵ h_x              + δ - γ * β / ϵ
    ] // βCollect];
gm_{a_ b_→c_}[β_] := β // sw_ab // hm_{ab→c} // tm_{ab→c};
B /: B[ω1_, Λ1_] B[ω2_, Λ2_] := B[ω1 * ω2, Λ1 + Λ2];
Rp_{a_ b_} := B[1, (X - 1) t_a h_b];
Rm_{a_ b_} := B[1, (X^-1 - 1) t_a h_b];
```

```
{β = B[ω, Sum[α_{10 i+j} t_i h_j, {i, {1, 2, 3}}, {j, {4, 5}}]],
  (β // tm_{12→1} // sw_14) == (β // sw_24 // sw_14 // tm_{12→1})}
```

$$\left\{\begin{pmatrix} \omega & h_4 & h_5 \\ t_1 & \alpha_{14} & \alpha_{15} \\ t_2 & \alpha_{24} & \alpha_{25} \\ t_3 & \alpha_{34} & \alpha_{35} \end{pmatrix}, \text{True}\right\}$$




Some testing




```
{Rm_51 Rm_62 Rp_34 // gm_{14→1} // gm_{25→2} // gm_{36→3},
  Rp_61 Rm_24 Rm_35 // gm_{14→1} // gm_{25→2} // gm_{36→3}}
```




$$\left\{\begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+X}{X} & 0 \\ t_3 & \frac{-1+X}{X} & -\frac{-1+X}{X} \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+X}{X} & 0 \\ t_3 & \frac{-1+X}{X} & -\frac{-1+X}{X} \end{pmatrix}\right\}$$

… divide and conquer!

```
β = Rm_{12,1} Rm_27 Rm_83 Rm_{4,11} Rp_{16,5} Rp_{6,13} Rp_{14,9} Rp_{10,15}
```

$8_{17}$

$$\begin{pmatrix} 1 & h_1 & h_3 & h_5 & h_7 & h_9 & h_{11} & h_{13} & h_{15} \\ t_2 & 0 & 0 & 0 & -\frac{-1+X}{X} & 0 & 0 & 0 & 0 \\ t_4 & 0 & 0 & 0 & 0 & 0 & -\frac{-1+X}{X} & 0 & 0 \\ t_6 & 0 & 0 & 0 & 0 & 0 & 0 & -1+X & 0 \\ t_8 & 0 & -\frac{-1+X}{X} & 0 & 0 & 0 & 0 & 0 & 0 \\ t_{10} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+X \\ t_{12} & -\frac{-1+X}{X} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ t_{14} & 0 & 0 & 0 & 0 & -1+X & 0 & 0 & 0 \\ t_{16} & 0 & 0 & -1+X & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

```
Do[β = β // gm_{1k→1}, {k, 2, 10}]; β
```

$8_{17}$, cont.

$$\begin{pmatrix} \frac{1}{X} & h_1 & h_{11} & h_{13} & h_{15} \\ t_1 & -\frac{(-1+X)(1+X)}{X} & -(-1+X)(1-X+X^2) & (-1+X)(1-X+X^2) & -1+X \\ t_{12} & -\frac{-1+X}{X} & 0 & 0 & 0 \\ t_{14} & -1+X & \frac{(-1+X)^2(1-X+X^2)}{X} & -\frac{(-1+X)^2(1-X+X^2)}{X} & 0 \\ t_{16} & \frac{-1+X}{X} & (-1+X)^2 & -\frac{(-1+X)^3}{X} & 0 \end{pmatrix}$$



James Waddell Alexander

```
Do[β = β // gm_{1k→1}, {k, 11, 16}]; β
```

$$\left(-\frac{1-4X+8X^2-11X^3+8X^4-4X^5+X^6}{X^3}\right)$$

A Partial To Do List. 1. Where does it *more simply* come from?
2. Remove all the denominators.
3. How do determinants arise in this context?
4. Understand links ("meta-conjugacy classes").
5. Find the "reality condition".
6. Do some "Algebraic Knot Theory".
7. Categorify.
8. Do the same in other natural quotients of the v/w-story.




trivial




ribbon



example



"God created the knots, all else in topology is the work of mortals."
Leopold Kronecker (modified)

www.katlas.org

Video and more at http://www.math.toronto.edu/~drorbn/Talks/Newton-1301/, see also .../Sheffield-130206/ and .../Regina-1206/